
Experiment #19  
A Light-Controlled Theremin 

 
This experiment demonstrates FREQOUT by creating a light-controlled Theremin (the first electronic 
musical instrument ever produced). 
 
Building The Circuit 
 

 
 
Note: Later versions of the StampWorks lab board come with a built-in audio amplifier.  Attach an 8-
ohm speaker to the output of the amplifier to get the best sound from this project. 
 
' ============================================================================== 
' 
'   File...... Ex19 - Theremin.BS2 
'   Purpose... Simple Digital Theremin 
'   Author.... Parallax 
'   E-mail.... stamptech@parallaxinc.com 
'   Started... 
'   Updated... 01 MAY 2002 
' 
'   {$STAMP BS2} 
' 
' ============================================================================== 
 
 
' ------------------------------------------------------------------------------ 
' Program Description 
' ------------------------------------------------------------------------------ 

' This program uses RCTIME with a photocell to create a light-controlled 
' theremin. 
 
 
' ------------------------------------------------------------------------------ 
' I/O Definitions 
' ------------------------------------------------------------------------------ 
 
Speaker         CON     0                       ' piezo speaker output 
PitchCtrl       CON     1                       ' pitch control (RCTIME) input 
 
 
' ------------------------------------------------------------------------------ 
' Constants 
' ------------------------------------------------------------------------------ 
 
Scale           CON     $0100                   ' divider for BS2/BS2e 
'Scale          CON     $0066                   ' divider for BS2sx 
'Scale          CON     $0073                   ' divider for BS2p 
 
Threshold       CON     200                     ' cutoff frequency to play 
 
 
' ------------------------------------------------------------------------------ 
' Variables 
' ------------------------------------------------------------------------------ 
 
tone            VAR     Word                    ' frequency output 
 
 
' ------------------------------------------------------------------------------ 
' Program Code 
' ------------------------------------------------------------------------------ 
 
Main: 
  HIGH PitchCtrl                                ' discharge cap 
  PAUSE 1                                       ' for 1 ms 
  RCTIME PitchCtrl, 1, tone                     ' read the light sensor 
  tone = tone */ Scale                          ' scale input 
 
  IF (tone < Threshold) THEN Main               ' skip for ambient light 
  FREQOUT Speaker, 25, tone                     ' output the tone 
  GOTO Main 
 
  END 

Behind The Scenes 
 
A Theremin is an interesting musical device used to create those weird, haunting sounds often heard 
in old horror movies. This version uses the light falling onto a photocell to create the output tone.   
 
Since the photocell is a resistive device, RCTIME can be used to read its value. FREQOUT is used to 
play the note. The constant, Threshold, is used to control the cutoff point of the Theremin. When 
the photocell reading falls below this value, no sound is played. This value should be adjusted to the 
point where the Theremin stops playing when the photocell is not covered in ambient light. 
 
Challenge 
 
Add a second RC circuit using a 10K pot instead of a photocell. Use this circuit to adjust the threshold 
value to varying light conditions. 


